
概要

複雑系を読み解く糸口としてハースト指数が有用である。C.Reiterをフォローしながら太陽

黒点の 500年間のデータでハースト指数とフラクタル時系列を試してみる。

1 ハーストと R/S解析

1.1 ハースト

ハースト（Harold Edwin Hurst 1880-1978 England)は Leicesterの近くで生まれ, Oxfordで物理

学を専攻。卒業後 3年間大学に残って教えていたが、1906年に Sir Henry Lyons (エジプト調査測

量局長官）に請われてエジプトに渡り、地磁気と天文観測に従事したが、間もなく気象と水理部門

も加わり、1915年に物理部門の責任者となった。当時はイギリスのエジプト統治の時代であった。

この頃、ナイル河の水理の管理プロジェクトが持ち上がり、計画に先立ちスーダンや東アフリカ

を含むナイル河と広範な支流ネットワークの降雨量と氾濫の解析が必要となり、ハーストをヘッド

とする物理部門がデータを集め解析に永年取り組んだ。

1946年に Hurst Black Simaika [Nail Basin]を刊行。

広範な長期にわたる治水計画の立案の基礎となりアスワンハイダムや Jonglei運河の計画に繋が

る。引退後も 88才までエジプト公共事業省の科学顧問を務めた。

このナイル河の水文解析から R/S解析とハースト指数を考案し、ハースト指数 0.5をランダムウ

オークとし、ナイル河のハースト指数は 0.72で長期記憶のあるデータであるとした。(LongTerm

Strage　 1965)

この理論はエンジニアリング部門ですぐに実用化され、数学ではマンデルブロートなどが注目

した。

かって NASAの通信トラフィックの解析が試みられ、金融面への応用も始まっている。

1.2 ランダムウオーク

ランダムウオークの生成アルゴリズムの例を示す。

• 乱数を 20個打ち出す (?.固定した乱数を生成）

20?.20

6 3 19 15 10 14 0 7 12 17 16 4 13 2 1 9 18 5 11 8

• 20の 2分の 1（10）との比較。

(-:20)> 20?.20 NB. -: is half

1 1 0 0 0 0 1 1 0 0 0 1 0 1 1 1 0 1 0 1

• 0 1を指標として 1 _1を選ぶ (ランダムウオーク）

((-:20)> 20?.20){1 _1

_1 _1 1 1 1 1 _1 _1 1 1 1 _1 1 _1 _1 _1 1 _1 1 _1

• 　累積

7{. <\((-:20)> 20?.20){1 _1　 NB.7個を表示

+--+-----+-------+---------+-----------+-------------+----------------+

|_1|_1 _1|_1 _1 1|_1 _1 1 1|_1 _1 1 1 1|_1 _1 1 1 1 1|_1 _1 1 1 1 1 _1|

+--+-----+-------+---------+-----------+-------------+----------------+

• 累積和 (box (¡)は他の関数で置き換え簡略化できる）

+/\((-:20)> 20?.20){1 _1

_1 _2 _1 0 1 2 1 0 1 2 3 2 3 2 1 0 1 0 1 0

• スクリプトに纏める
rw=: 3 : ’+/\((-:y)>y?y){1 _1’

ランダムウオークは重心が 0.5からほんの少し傾いても明確な上昇または下降のトレンドが

入る。

図 1 RW 0.48 0.5 0.52

plot 0.48 0.5 0.52 rwp 1000

次の rwpは多数のウエイトを同時に計算、表示できる。

(3#0.5) rwp 1000等とすれば、複数のランダムウオーク（0.5）を生成できる。

rwp=:4 : 0

NB. Random walk with many probability

NB. Usage: plot 0.3 0.5 0.6 rwp 1000

NB. or 0.5

;("1),. +/\ L:0({(x * y)> y ? (#x)#y){(L:0)_1 1

)

1.3 ブラウン運動の生成

ブラウン運動は平均と分散または標準偏差を指定した正規乱数から生成できる。dt = 0.01,dx =

±
√

0.01のように時間間隔とジャンプの幅を微少にして連続関数を擬する。

時間の幅 dt,ジャンプの幅 dx= σ
√

dtとする

P(Xn = dx) =
1
2
+
µ

2σ2
dx

P(Xn = −dx) =
1
2
− µ

2σ2
dx

Xnの平均と分散

E(Xn) = dx×
(
1
2
+
µ

2σ2
dx

)
− dx×

(
1
2
+
µ

2σ2
dx

)
= dx× µ

2σ2
dx= µdt

V(Xn) = E(X2
n) − (E(Xn))2 = σ2dt− µ2(dt)2

Xnから Ynを生成する

Y0 = 0,Yn = Y1 + Y2 + · · · + Yn

正規乱数の生成 scriptは個数のみ指定できるものが多いが、次の N.Thomsonの rhoは平均と標

準偏差のパラメータも用いることができる。

図 2 正規乱数

plot 0.1 count0 rno 0, 0.5, 1,10000

ブラウン運動のサンプルパスは、所定の正規乱数を生成した後にランダムウオークと同じように

+/\で生成できる。

] a=. rn0 0 0.5 10

a,. +/\ a

1.16829 1.16829

_0.0981223 1.07017

_1.09988 _0.0297081

0.968845 0.939137

_0.0999985 0.839138

_0.520794 0.318345

0.338691 0.657035

_1.09443 _0.437395

_0.268586 _0.70598

_0.374875 _1.08086

plot((i.1000)%100); +/\ rno 0 0.1 1000

pd ’eps /temp/bwn_1.eps’

4



0 1 2 3 4 5 6 7 8 9 10

-3

-2.5

-2

-1.5

-1

-0.5

0

0.5

1

1.5

図 3 Brown運動 µ = 0σ = 0.1

1.4 R/S解析 ·長い記憶の検定

長い記憶（long memory)とは自己相関が緩慢に小さくなっていく過程である。長い記憶を持つ

かどうかの検定として、ハーストにより開発された R/S統計量が知られている。

Rt,N = max1≤k≤N(
∑k

i=1(xt+i − x̄t,N)) −min1≤k≤N(
∑k

i=1(xt+i − x̄t,N))

St,Nは標準偏差
Rt,N

St,N
' Q̄t,N ' cNH

MAX

MIN
R

• 正規乱数を指定個数打ち出す　 randsnは正規乱数生成

] a=. randsn 5

_1.18731 0.0238542 0.64544 1.30989 0.1113

• 偏差と累積和を求める

a,.((- mean)a),. +/\ (- mean) a

正規乱数 deviation 累積和 ccd

_1.18731 _1.36794 _1.36794

0.0238542 _0.15678 _1.52472

0.64544 0.464806 _1.05992

1.30989 1.12925 0.0693349

0.1113 _0.0693349 1.38778e_17

ccd=:[: +/\ (- mean)

• 累積和の最大と最小を取り出し、その差を求める。

r0 a

1.59406

r0=:(>./ - <./)@ccd

• 最大と最小の差を標準偏差で割り、単位の影響を除去する

q0 a

1.93538

q0=:r0 % sd NB. sd is standard deviation

• *この計算を次の分割した窓毎に行う

1.4.1 リスケール

ここではガリレオの時代より少し前に開始された太陽黒点の 500年間のデータを用いる。最初

の 400年（1500-1899)はC.Reiterによる。ベルギーの S IDIに 1800年以降のデータが上がってい

る。今回の解析では双方のデータをマージした。(n = 508)

• データの tを分割する窓 (timespan)を設ける。*1

span　 NB. 時間の窓

span=: 8 16 32 64 127 254 508 NB. n=508

• span毎の標準偏差の最大値と最小値の差を求める。

ナイル河の水位の時間分割での標準偏差の最大値と最小値に相当する

*1 waveletの様に 2nで分割する方法もある。経過は 2nによる。実解析はデータ総数の半分の数を最大とする。

図 4 時間の窓の構成

窓のスケール毎に差を平均する（窓が小さいほど差も小さくなる）

|. span,. span rs_test sp0

508 66.9287 NB. 1/1

254 45.8116 NB. 1/2

127 27.489 NB. 1/4

64 14.4251

32 8.13549

16 5.93442

8 3.49794

• spanを x（説明変数）として両対数 (自然対数）で回帰する

(ˆ. span rs sp1) %. 1,. ˆ. span

_0.304521 0.731273

f = −0.304521+ 0.731273lnx

0.73がハースト指数となる。

calc hurstによるハースト指数の計算。

xの取り方が経過の簡略法と多少異なる。H = 0.769

calc_hurst sp1

_0.479422 0.769562

1.4.2 ハースト指数のスクリプト

rs_test=:([: mean ,:˜@[q0 ;._3])"0 1 NB. same

calc_hurst=: 3 : 0

NB. calc_hurst randsn 1000

N=: 2+i.<.-:#y

RS=. N rs_test y NB. R/S static

(ˆ. RS)%. 1,. ˆ. N

)

窓は nの半分程度までに設定する。

N=. 2+i.<.-:#y

1.5 ハースト指数とランダムウオーク

ハースト指数は 0.5でランダムウオークであり、0.5より小さいデータは長期の記憶を持ってい

ない。0.5を超えて 1に近づくほど長期記憶を有する。

1.5.1 ハースト指数の計算・経過と解説

1.5.2 rs testの作成

• cut 3を用いて 5ずつ区切る。*2

(5,:5)<;.(_3) 10 ?. 10

+---------+---------+

|6 9 1 4 0|2 3 8 7 5|

+---------+---------+

• box(<)は動詞（関数）に置換えることができる

(5,:5)q0 ;.(_3) 10 ?. 10

2.13003 2.19265

• ,:˜でカットの関数を整理する

5 (,:˜@[q0 ;._3]) 10 ?. 10

2.13003 2.19265

• 平均　（mean=. +/%#)を付加

*2 3は cutの方法のパラメータ。指標ではなく数（左引数）でカットする箇所を指定する

5 ([: mean ,:˜@[q0 ;._3]) 10 ?. 10

2.16134

• 関数に纏める

rs_test=: ([: mean ,:˜@[q0 ;._3])

5 rs_test 10?.10

2.16134

• 多くの区切りを左引数にとって同時に計算する

4 5 rs_test"(0 1) 10?.10

1.66327 2.16134

1000個の乱数のハースト指数の例。(窓は自動生成)

calc_hurst randsn 1000

_0.294702 0.618119

フラクタルで用いるマンデルブロート指数は

2− H である。時系列データではハースト指数の方がが求めやすい。

2 フラクタル時系列

2.1 太陽黒点とハースト指数

データは

1500-1699は C.Reiter

1700-2008は S IDC(Solar Influence Data Analsys

Center,Bergium)によった。

1650-1700と 1800-1830頃に太陽黒点の少ない

時期がある。

17世紀のミニ氷河の襲来期にはテムズ河でス

ケートが出来たと報告されている。

2.2 自己相関 Autocorreration

自己相関は過去との相関を一纏めにしてあらわす。

ACF(k) =
Σn

t=k−1(Yt − Ȳ)(Yt−k − Ȳ)

Σn
t=1(Yt − Ȳ)2

• EXAMPLE　 i.10 (0から 9)

• 次の組み合わせを作成する。1本のデータを copyして 2本の紐とする。1ステップずらし

て手でもって、はみ出た上と下を切って同じ長さとする。1ステップずつ繰り返す。*3

({@>1+i.8) ,. }.}: ( (<\.a),.|.@(<\) a=. i.10)

+-+-----------------+-----------------+

|1|1 2 3 4 5 6 7 8 9|0 1 2 3 4 5 6 7 8|

+-+-----------------+-----------------+

|2|2 3 4 5 6 7 8 9 |0 1 2 3 4 5 6 7 |

+-+-----------------+-----------------+

|3|3 4 5 6 7 8 9 |0 1 2 3 4 5 6 |

+-+-----------------+-----------------+

|4|4 5 6 7 8 9 |0 1 2 3 4 5 |

+-+-----------------+-----------------+

|5|5 6 7 8 9 |0 1 2 3 4 |

+-+-----------------+-----------------+

|6|6 7 8 9 |0 1 2 3 |

+-+-----------------+-----------------+

|7|7 8 9 |0 1 2 |

+-+-----------------+-----------------+

|8|8 9 |0 1 |

+-+-----------------+-----------------+

• 平均（総データの）を引いて掛け合わせ列毎に合計する

tmp,. +/ L:0 tmp=. */ L:0 ,. <"2 > ((<\. a),.|. <\ a) - L:0 mean a

+-----------------------------------------------------+------+

|20.25 12.25 6.25 2.25 0.25 0.25 2.25 6.25 12.25 20.25|82.5 |

+-----------------------------------------------------+------+

|15.75 8.75 3.75 0.75 _0.25 0.75 3.75 8.75 15.75 0 |57.75 |

*3 最初は同一、最後は一個。経過の解説では省略してある

+-----------------------------------------------------+------+

|11.25 5.25 1.25 _0.75 _0.75 1.25 5.25 11.25 0 0 |34 |

+-----------------------------------------------------+------+

|6.75 1.75 _1.25 _2.25 _1.25 1.75 6.75 0 0 0 |12.25 |

+-----------------------------------------------------+------+

|2.25 _1.75 _3.75 _3.75 _1.75 2.25 0 0 0 0 |_6.5 |

+-----------------------------------------------------+------+

|_2.25 _5.25 _6.25 _5.25 _2.25 0 0 0 0 0 |_21.25|

+-----------------------------------------------------+------+

|_6.75 _8.75 _8.75 _6.75 0 0 0 0 0 0 |_31 |

+-----------------------------------------------------+------+

|_11.25 _12.25 _11.25 0 0 0 0 0 0 0 |_34.75|

+-----------------------------------------------------+------+

|_15.75 _15.75 0 0 0 0 0 0 0 0 |_31.5 |

+-----------------------------------------------------+------+

|_20.25 0 0 0 0 0 0 0 0 0 |_20.25|

+-----------------------------------------------------+------+

• 偏差の 2乗和を求める

(+/@:ˆ&2@(- +/%#) ) a=. i.10

82.5

• 2乗和で割り ac f を得る

acfx a

1 0.7 0.412121 0.148485 _0.0787879 _0.257576 _0.375758 _0.421212 _0.381818 _0.245455

• Script

acfx=:3 : ’((+/@:*/)"2>((<\.y),.|.<\y) - L:0 mean y) % +/@:(ˆ&2)@dev y’

太陽黒点 (1500-1008)の自己相関係数の最初の

15年間

15{.(>:i. # a),.a=.acf sp0

1 1

2 0.834458

3 0.486417

4 0.094268

5 _0.211901

6 _0.364714

7 _0.33541

8 _0.138826

9 0.159475

10 0.467301

11 0.662376 NB. runner-up

12 0.676837 NB. max

13 0.516813

14 0.254852

15 _0.0130849

508年間の自己相関係数の推移

12の自己相関係数が最大である。「太陽黒点には

11年周期の増加と減少の周期があり、22年が 1

サイクルである」と言われることと符合する。

2.3 ハースト指数とフラクタル次元

ハースト指数とフラクタル次元は 2− H で表すことができる。

そこで、ハースト指数と正規乱数を用いて hwalkを作成し、ハースト指数とブラウン運動の関係を

図で見られるようにする。

√
1− 22H−2

(
1
2h

)k+1

スクリプトは C.Reiterに依った。

√
1− 22H−2

(
1
2h

)k+1

interp=: (}. + }:)@:(2: # -:)

osz=: %:@-.@(2&ˆ)@+:@<:@[

sz0=: osz * %@+:@<:@#@] ˆ [

randunif=: (?%<:)@:($&2147483647) : ({.@[+({:-{.)@[*$:@]) NB. uniform rand

randsn=: cos@+:@o.@randunif * %:@-@+:@ˆ.@randunif NB. normal rand

randadd=: ] + sz0 * randsn@$@]

hwalk=: 4 : 0

x ([ randadd interp@]) ˆ: y 0,(x osz 1)* randsn 1

)

0 50 100 150 200 250

-2

-1

0

1

2

3

図 5 hwalk,H=0.1

0 50 100 150 200 250

-1

-0.8

-0.6

-0.4

-0.2

0

0.2

0.4

0.6

図 6 Hwalk,H=0.5

図 7 hwalk,H=0.9

2.4 R/S解析と太陽黒点

a=. ; }.|: sp=. > jread ..

plot calc_rs_sub a

図 8 lnR/S vs ln N

太陽黒点の周辺では爆発現象が多く起こり、黒点

が多いと太陽の活動が活発である。

この図は太陽黒点の 508年間のデータをハース

トの R/S解析の両対数を Y に R/Sの値、X に N=

スケールのサイズを取ったものである。

ln2.5付近に回折点がある。ln2.5を戻すと

ˆ2.5

12.1825

となる。

そこで、ハースト指数を全てのスケールで求めると 0.77

calc_hurst a

_0.479422 0.769562

ln2.5でマスクを掛け、12以下のスケールで求めると 0.74となる。長期記憶を有している。

2.5 calc_rs_masked a

_0.535591 0.743711

マスクを掛けた回帰は次の様になる。

0.5 1 1.5 2 2.5 3 3.5 4 4.5 5 5.5

0

0.5

1

1.5

2

2.5

3

3.5 rs_sunspots
reg

図 9 lnR/S masked ln2.5

2.4.1 Script

mean=:+/%#

diff=: }. - }: NB. 2 stage Sabun

ccd=: [: +/\ (- mean) NB. same avobe

sd=: [: %: # %˜ (+/@: *:@ (- mean))

r0=:((>./)-<./ )@ccd NB. max - min

q0=: r0 % sd

rs=: ([: mean ,:˜@[ q0 ;._3 ]) "0 1 NB. R/S static

rs_test=:([: mean ,:˜@[q0 ;._3])"0 1 NB. same

calc_hurst=: 3 : 0

NB. calc_hurst sunspots

N=: 2+i.<.-:#y

RS=. N rs_test y NB. R/S static

(ˆ. RS)%. 1,. ˆ. N

)

NB. -------------------------

calc_rs_sub=: 3 : 0

X0=. diff diff y

N=. 2+i. <. -: # X0 NB. komado

RS=. N rs_test X0 NB. rs

(ˆ. N);ˆ. RS

15



)

calc_rs_masked=: 4 : 0

X0=. diff diff y

N=. 2+i. <. -: # X0 NB. komado

RS=. N rs_test X0 NB. rs

MASK=. N <: ˆ x NB. 1 1 1.. 0 0 0....

(ˆ. MASK # RS)%. 1,. ˆ. MASK # N

)

plot_rs=: 4 : 0

NB. find 2.5 by own eye at x axis//

’N RS’=: calc_rs_sub y

pd ’reset’

pd ’key rs_sunspots reg’

pd N; RS,: (1,. N) +/ . * f=. x calc_rs_masked y

pd ’show’

)

2.5 フラクタル時系列と予測

C.Reiterが非常にデリケートだがカオスが顔を見せて魅力的であるというフラクタル予測を、

C.Reiterのスクリプトとともに鑑賞しよう。

m− histories,k（m-histriesの周辺の数）の 2個のパラメーターが重要である。

2.5.1 エノンのカオスデータ

hen=:3 : ’y,1+(0.3*_2{y)+_1.4**: _1{y’

plot hen ˆ:(400)]0 0

ref　一個ずらしたデータの組み合わせ

2]\ i.5

0 1

1 2

2 3

3 4

図 10 henon 100

距離の差の幾何平均

1 2 3 dist 3 4 2

3

dist

%:@(+/)@:*:@:-"1

解説

(a=.1 2 3,: 3 4 2),

(-/ a),:ˆ&2 -/ a

1 2 3 NB. A

3 4 2 NB B

---------

_2 _2 1 NB. -/ A,B

4 4 1 NB. ˆ&2

------------------

+/ 4 4 1 is 9 NB. sum

%: 9 is 3 NB. square

data=. hen ˆ:(400) ] 0 0

m=2 k=5 size=: 350とする

ref=2 ]\ size{. data

sizeの少し上のデータを求める

refとの距離の幾何平均を計算する

] x=. (350+1+i.m) {data

0.364627 0.564313

5{. ref dist x

0.671865 0.568134 1.15482 0.920043 1.48647

アドレスを保存しておく

]j=. (k=.5){. /: ref dist x

230 68 196 267 91

x,yを求める

(m+j),. (m+j){data

adress data

------------

232 0.672215

70 0.695775

198 0.722901

269 0.600377

93 0.588537

1,. j{ref

1 0.367706 0.559398

1 0.383374 0.547225

1 0.398893 0.532358

1 0.32659 0.596178

1 0.319012 0.601882

回帰する

] coef=. ((m+j){data) %. 1,. j{ref

1.23331 0.516729 _1.34456

f = .1.23331+ 0.516729x− 1.34456x2

1期の予測

(>:size+i.m){data

0.364627 0.564313

coef +/ . * 1, (>:size+i.m){data

0.662971

20 ステップの予測計算（non-update

type) param=. 2 5 350

5{. a,. (a=.param,"1 0]>:i.20)

fracpred data0

m k siz T result

2 5 350 1 38.0975

2 5 350 2 25.9315

2 5 350 3 14.0684

2 5 350 4 6.58168

2 5 350 5 _2.46848

• 　 update　 type (cの推計用）

• 反復計算エンジン f ps1 m− histryのため mを更新する

2 5 350 fps1 data;(>:350+i.m){data

0.662971

• Script

fps1=:4 : 0("1)

NB. Usage: 2 5 350 fps1 data0

’m0 k0 sz0 ’=. x

’y0 x0’ =. y

NB. x0=. (sz0+1+i.m0){y0

ref=. m0 ]\ sz0 {. y0

j0=. k0{./: ref dist x0

coef=. ((m0+j0){y0) %. 1,.j0{ref

coef +/ . * 1,x0

)

• 連続計算のエンジン f ps2

data0と yは分離する。非常にデリケートである

fps2=:3 : ’ param fps1 data0;y’

param=. 2 5 350

• ドライブする (200step)

a=. (}.@, fps2) ˆ:(>:i.200) (>:size+i.m){data

plot }.{. |: a

図 11 timehenon,m=2

• param=. 3 7 350　で

図 12 timehenon,m=3

どちらかの行を用いればよい

2.5.2 サンスポットデータで

• データ
sp0=. ;{:|: sp NB. 508 years sunspots data (jread ...)

fps1用に data0=: sp0　と置く。(data0は必ずグローバル定義が必要)

• パラメータ
mデータを分割する単位 (m-histry)　 ARの次数に相当か

k mの近傍のデータ、mの 2倍より大きめ

sz n-G-m程度

Gは　予測年数　 umber of steps in our goal forecast

• 計算出力
– a0予測部分のオリジナルデータ

– b0 mを固定

– c0mを逐次更新したフラクタル予想

– a0,b0,:c0の形式

• Script

pred_frac=: 4 : 0

NB. Usage:8 20 460 35 pred_frac sunspots

NB. x is m k sz G

NB. or/ y is ; }. |: sp0

’M0 K0 SZ0 G0’=: x

data0=: y NB. should global definition

NB. -----calc----------------------

a0=. (>: SZ0+M0+i.G0){ y

b0=. ((M0,K0,SZ0),("1 0)] >:i.G0) fracpred y

c0=.(<:M0)}.{.|:(}.@, fps2)ˆ:(>:i.G0+<:M0)(>:SZ0+i.M0){y

a0,b0,:c0

)

• Example

8 20 460 35 pred_frac ; }.|: sp0

8 20 460 35 plot_frac ; }.|: sp0

図は m=8 k=20 sz=460 G=35としたフラクタル予想 (浅緑）である。

図 13 fractal prediction m=8,k=20

パラメータに非常にデリケートであるが魅力的な結果をもたらす。更に実例を積み重ねて検証す

る必要がある。

